apply plugin: 'it.gianluz.capsule'

if (JavaVersion.current().compareTo(JavaVersion.VERSION_11) >= 0) {

    apply plugin: 'org.openjfx.javafxplugin'

    javafx {
        modules = [ 'javafx.controls', 'javafx.web' ]
    }
}

dependencies {
    compile project(':extensions:swt')
    compile project(':extensions:treetable')
    compile project(':extensions:jfreechart')
    compile project(':extensions:javafx')
    compile project(':extensions:nachocalendar')
    compile project(':extensions:jgoodiesforms')
    compile files('lib/http.jar')
    compile 'org.kohsuke:github-api:1.111'
}

/**
 * Issues-Browser
 */
task issuesbrowserjar(type: FatCapsule) {
    destinationDir = file("$rootDir/$targetDir")
    archiveName = "glazedlists-demo.jar"

    applicationClass 'com.publicobject.issuesbrowser.swing.IssuesBrowser'

    capsuleManifest {
        applicationId = 'com.publicobject.issuesbrowser.swing.IssuesBrowser'
        applicationName = 'IssuesBrowser'
        minJavaVersion = '1.8.0'
    }
}

/**
 * XML-Browser
 */
task xmlbrowserjar(type: FatCapsule) {
    destinationDir = file("$rootDir/$targetDir")
    archiveName = "xmlbrowser.jar"

    applicationClass 'com.publicobject.xmlbrowser.XmlBrowser'

    capsuleManifest {
        applicationId = 'com.publicobject.xmlbrowser.XmlBrowser'
        applicationName = 'XmlBrowser'
        minJavaVersion = '1.8.0'
    }
}